NAG Fortran Library Routine Document

F08BVF (ZTZRZF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08BVF (ZTZRZF) reduces the m by n ($m \leq n$) complex upper trapezoidal matrix A to upper triangular form by means of unitary transformations.

2 Specification

```
SUBROUTINE F08BVF (M, N, A, LDA, TAU, WORK, LWORK, INFO)

INTEGER             M, N, LDA, LWORK, INFO
complex*16          A(LDA,*), TAU(*), WORK(*)
```

The routine may be called by its LAPACK name ***ztzrzf***.

3 Description

The m by n ($m \leq n$) complex upper trapezoidal matrix A given by

$$A = \begin{pmatrix} R_1 & R_2 \end{pmatrix},$$

where R_1 is an m by m upper triangular matrix and R_2 is an m by $(n-m)$ matrix, is factorized as

$$A = \begin{pmatrix} R & 0 \end{pmatrix} Z,$$

where R is also an m by m upper triangular matrix and Z is an n by n unitary matrix.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $M \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $N \geq 0$.

3: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the leading M by N upper trapezoidal part of the array A must contain the matrix to be factorized.

On exit: the leading M by M upper triangular part of A contains the upper triangular matrix R, and elements $M+1$ to N of the first M rows of A, with the array TAU, represent the unitary matrix Z as a product of m elementary reflectors.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08BVF (ZTZRZF) is called.

Constraint: LDA $\geq \max(1, \mathrm{M})$.

5: TAU($*$) – ***complex*16*** array *Output*

Note: the dimension of the array TAU must be at least $\max(1, \mathrm{M})$.

On exit: the scalar factors of the elementary reflectors.

6: WORK($*$) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.

7: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08BVF (ZTZRZF) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK $\geq \mathrm{M} \times nb$, where nb is the optimal ***block size***.

Constraint: LWORK $\geq \max(1, \mathrm{M})$ or LWORK $= -1$.

8: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed factorization is the exact factorization of a nearby matrix $A + E$, where

$$\|E\|_2 = O\,\epsilon\|A\|_2$$

and ϵ is the ***machine precision***.

8 Further Comments

The total number of floating point operations is approximately $16m^2(n - m)$.

The real analogue of this routine is F08BHF (DTZRZF).

9 Example

This example solves the linear least squares problems

$$\min_x \left\|b_j - Ax_j\right\|_2, \quad j = 1, 2$$

for the minimum norm solutions x_1 and x_2, where b_j is the jth column of the matrix B,

$$A = \begin{pmatrix} 0.47-0.34i & -0.40+0.54i & 0.60+0.01i & 0.80-1.02i \\ -0.32-0.23i & -0.05+0.20i & -0.26-0.44i & -0.43+0.17i \\ 0.35-0.60i & -0.52-0.34i & 0.87-0.11i & -0.34-0.09i \\ 0.89+0.71i & -0.45-0.45i & -0.02-0.57i & 1.14-0.78i \\ -0.19+0.06i & 0.11-0.85i & 1.44+0.80i & 0.07+1.14i \end{pmatrix} \quad \text{and}$$

$$B = \begin{pmatrix} -1.08-2.59i & 2.22+2.35i \\ -2.61-1.49i & 1.62-1.48i \\ 3.13-3.61i & 1.65+3.43i \\ 7.33-8.01i & -0.98+3.08i \\ 9.12+7.63i & -2.84+2.78i \end{pmatrix}.$$

The solution is obtained by first obtaining a QR factorization with column pivoting of the matrix A, and then the RZ factorization of the leading k by k part of R is computed, where k is the estimated rank of A. A tolerance of 0.01 is used to estimate the rank of A from the upper triangular factor, R.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

```
*     F08BVF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NMAX, NRHSMX
      PARAMETER        (MMAX=8,NB=64,NMAX=8,NRHSMX=2)
      INTEGER          LDA, LDB, LWORK
      PARAMETER        (LDA=MMAX,LDB=MMAX,LWORK=(NMAX+1)*NB)
      COMPLEX *16      ONE, ZERO
      PARAMETER        (ONE=(1.0D0,0.0D0),ZERO=(0.0D0,0.0D0))
*     .. Local Scalars ..
      DOUBLE PRECISION TOL
      INTEGER          I, IFAIL, INFO, J, K, M, N, NRHS
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), B(LDB,NRHSMX), TAU(NMAX),
     +                 WORK(LWORK)
      DOUBLE PRECISION RNORM(NMAX), RWORK(2*NMAX)
      INTEGER          JPVT(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Functions ..
      DOUBLE PRECISION DZNRM2
      EXTERNAL         DZNRM2
*     .. External Subroutines ..
      EXTERNAL         F06DBF, F06THF, X04DBF, ZCOPY, ZGEQP3, ZTRSM,
     +                 ZTZRZF, ZUNMQR, ZUNMRZ
*     .. Intrinsic Functions ..
      INTRINSIC        ABS
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08BVF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N, NRHS
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.GE.N .AND. NRHS.LE.NRHSMX)
     +    THEN
*
*        Read A and B from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,M)
*
*        Initialize JPVT to be zero so that all columns are free
*
```

```
         CALL F06DBF(N,0,JPVT,1)
*
*        Compute the QR factorization of A with column pivoting as
*        A = Q*(R11 R12)*(P**T)
*              ( 0  R22)
*
         CALL ZGEQP3(M,N,A,LDA,JPVT,TAU,WORK,LWORK,RWORK,INFO)
*
*        Compute C = (C1) = (Q**H)*B, storing the result in B
*                    (C2)
*
         CALL ZUNMQR('Left','Conjugate transpose',M,NRHS,N,A,LDA,TAU,B,
     +               LDB,WORK,LWORK,INFO)
*
*        Choose TOL to reflect the relative accuracy of the input data
*
         TOL = 0.01D0
*
*        Determine and print the rank, K, of R relative to TOL
*
         DO 20 K = 1, N
            IF (ABS(A(K,K)).LE.TOL*ABS(A(1,1))) GO TO 40
   20    CONTINUE
   40    K = K - 1
*
         WRITE (NOUT,*) 'Tolerance used to estimate the rank of A'
         WRITE (NOUT,99999) TOL
         WRITE (NOUT,*) 'Estimated rank of A'
         WRITE (NOUT,99998) K
         WRITE (NOUT,*)
*
*       Compute the RZ factorization of the K by K part of R as
*       (R1 R2) = (T 0)*Z
*
         CALL ZTZRZF(K,N,A,LDA,TAU,WORK,LWORK,INFO)
*
*        Compute least-squares solutions of triangular problems by
*        back substitution in T*Y1 = C1, storing the result in B
*
         CALL ZTRSM('Left','Upper','No transpose','Non-Unit',K,NRHS,ONE,
     +              A,LDA,B,LDB)
*
*        Compute estimates of the square roots of the residual sums of
*        squares (2-norm of each of the columns of C2)
*
         DO 60 J = 1, NRHS
            RNORM(J) = DZNRM2(M-K,B(K+1,J),1)
   60    CONTINUE
*
*        Set the remaining elements of the solutions to zero (to give
*        the minimum-norm solutions), Y2 = 0
*
         CALL F06THF('General',N-K,NRHS,ZERO,ZERO,B(K+1,1),LDB)
*
*        Form W = (Z**H)*Y
*
         CALL ZUNMRZ('Left','Conjugate transpose',N,NRHS,K,N-K,A,LDA,
     +               TAU,B,LDB,WORK,LWORK,INFO)
*
*        Permute the least-squares solutions stored in B to give X = P*W
*
         DO 100 J = 1, NRHS
            DO 80 I = 1, N
               WORK(JPVT(I)) = B(I,J)
   80       CONTINUE
            CALL ZCOPY(N,WORK,1,B(1,J),1)
  100    CONTINUE
*
*        Print least-squares solutions
*
         IFAIL = 0
```

```
         CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
     +               'Least-squares solution(s)','Integer',RLABS,
     +               'Integer',CLABS,80,0,IFAIL)
*
*        Print the square roots of the residual sums of squares
*
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     'Square root(s) of the residual sum(s) of squares'
         WRITE (NOUT,99999) (RNORM(J),J=1,NRHS)
      ELSE
         WRITE (NOUT,*)
     +     'One or more of MMAX, NMAX and NRHSMX is too small, ',
     +     'and/or M.LT.N'
      END IF
      STOP
*
99999 FORMAT (3X,1P,7E11.2)
99998 FORMAT (1X,I6)
      END
```

9.2 Program Data

```
F08BVF Example Program Data

   5            4             2                 :Values of M, N and NRHS

 ( 0.47,-0.34) (-0.40, 0.54) ( 0.60, 0.01) ( 0.80,-1.02)
 (-0.32,-0.23) (-0.05, 0.20) (-0.26,-0.44) (-0.43, 0.17)
 ( 0.35,-0.60) (-0.52,-0.34) ( 0.87,-0.11) (-0.34,-0.09)
 ( 0.89, 0.71) (-0.45,-0.45) (-0.02,-0.57) ( 1.14,-0.78)
 (-0.19, 0.06) ( 0.11,-0.85) ( 1.44, 0.80) ( 0.07, 1.14) :End of matrix A

 (-1.08,-2.59) ( 2.22, 2.35)
 (-2.61,-1.49) ( 1.62,-1.48)
 ( 3.13,-3.61) ( 1.65, 3.43)
 ( 7.33,-8.01) (-0.98, 3.08)
 ( 9.12, 7.63) (-2.84, 2.78)                            :End of matrix B
```

9.3 Program Results

```
 F08BVF Example Program Results

 Tolerance used to estimate the rank of A
      1.00E-02
 Estimated rank of A
      3

 Least-squares solution(s)
                   1                2
 1  ( 1.1669,-3.3224) (-0.5023, 1.8323)
 2  ( 1.3486, 5.5027) (-1.4418,-1.6465)
 3  ( 4.1764, 2.3435) ( 0.2908, 1.4900)
 4  ( 0.6467, 0.0107) (-0.2453, 0.3951)

 Square root(s) of the residual sum(s) of squares
      2.51E-01   8.10E-02
```